

Práctica 1.4.

Grafcet lineal

El contenido difundido o publicado compromete exclusivamente a los autores.

Proyectos de innovación: CÉLULA ROBOTIZADA



INDICE

1. Bloque de ejercicios para familiarizarse con el código Nivel 1. (Entradas, Salidas, Set, Reset, Temporización).....	3
<hr/>	
1.1.- ENUNCIADO	3
1.2.- Conocimientos previos	3
1.3.- Descripción de las entradas, salidas y instrucciones a utilizar	4
1.4.- Evolución gráfica por etapas.....	5
1.5.- Vista para la simulación.....	5
1.6.- Programa a realizar.....	6



1. BLOQUE DE EJERCICIOS PARA FAMILIARIZARSE CON EL CÓDIGO NIVEL 1. (ENTRADAS, SALIDAS, SET, RESET, TEMPORIZACIÓN)

1.1.- ENUNCIADO.

Se genera un grafset de cinco etapas, si se pulsa marcha se encenderá el indicador luminoso de dicho pulsador, pasados tres segundos saldrá el cilindro de servir muelles, una vez en esta posición si se pulsa paro y el sensor de cilindro extendido está activado, el cilindro entrará y el indicador luminoso se apagará.

1.2.- Conocimientos previos.

A la hora de hacer un Grafset habría que indicar que las salidas que activan cada etapa tienen que ir inmediatamente seguidas a las instrucciones de activación de la etapa. Para evitar que se cambie de etapa sin activar la salida.

Ejemplo:

Si se está en la Etapa 1 y se pulsa el S1, entonces se desactiva la Etapa 1 y se Activa la Etapa 2. Si se está en la Etapa 2 activar la Salida Q1.

```
If E1=1 and S1=1 then
    E1=0
    E2=1
endif
```

```
If E2=1 then Q1=1
```

Si se activa otra etapa antes de activar la salida se puede producir la situación de que no se active dicha salida:

```
If E1=1 and S1=1 then
    E1=0
    E2=1
Endif
```

```
If E2=1 and S1=0 then
    E2=0
    E3=1
endif
```

```
If E2=1 then Q1=1
```





Esto sucederá porque se ha cambiado de etapa y todavía no se ha activado la salida, de modo que cuando llega el momento de comprobar la salida, la etapa que activa dicha salida ya no está activa.

1.3.- Descripción de las entradas, salidas y instrucciones a utilizar.

Nombre	Descripción	Dirección	Función
S1	Pulsador de Marcha	Bit 3	Entrada
S2	Pulsador de Paro	Bit 4	Entrada
FC1	Final de carrera Cilindro 1 Dentro	Bit 8	Entrada
FC2	Final de carrera Cilindro 1 Fuera	Bit 9	Entrada
CIL1	Cilindro expendedor de muelles	Bit 8	Salida
L1	Indicador Luminoso Pulsador Marcha	Bit 0	Salida

ATENCIÓN: El cilindro expendedor de muelles estará fuera por defecto, es decir, al activar la bobina de la electroválvula, este cilindro entrará, y al desactivarla saldrá.

1.4.- Evolución gráfica por etapas. 1.5.- Vista para la simulación.





1.6.- Programa a realizar.

El programa queda con el siguiente aspecto:

```
'-----  
'Bloque de ejercicios para familiarizarse con el código Nivel 1.  
'(Entradas, Salidas, Set, Reset, Temporización)  
'Ejercicio 4  
'  
'Se genera un grafcet de cinco etapas, si se pulsa marcha se encenderá  
'el indicador luminoso de dicho pulsador, pasados tres segundos saldrá  
'el cilindro de servir muelles, una vez en esta posición si se pulsa  
'paro y el sensor de cilindro extendido está activado, el cilindro  
'entrará y el indicador luminoso se apagará.  
'-----  
  
'-----  
'Se definen las variables que se van a utilizar para las entradas y  
'las salidas.  
'-----  
  
DEF IO S1 = BIT, 3   'Se define el pulsador de marcha como entrada  
DEF IO S2 = BIT, 4   'Se define el pulsador de paro como entrada  
DEF IO FC1 = BIT, 8  'Se define el final de carrera de cilindro dentro como  
                     entrada.  
DEF IO FC2 = BIT, 9  'Se define el final de carrera de cilindro fuera como  
                     entrada  
DEF IO L1 = BIT, 0   'Se define el indicador luminoso del pulsador de marcha  
                     como salida  
DEF IO CIL1 = BIT, 8 'Se define el cilindro expendedor de muelles como salida  
  
'-----  
'Se definen las variables que se van a utilizar como etapas.  
'Se definirán como enteros, pero nosotros las utilizaremos con 1 o 0,  
'como booleanas.  
'-----  
  
Def inte E1  
Def inte E2  
Def inte E3  
Def inte E4  
Def inte E5  
  
'-----  
'Se definen las variables internas.  
'-----  
  
Def inte Temp  
  
'-----  
'Programa  
'-----  
  
'Se inicializa la variable Temporizador a 0  
Temp=0  
'Se asegura que el cilindro esté dentro  
CIL1=1
```



```
'Se declara la etiqueta de salto al inicio del programa
*Inicio

'Al inicio están todas las Etapas a 0, por lo tanto, pondremos la condición,
'si están todas las etapas a 0, pondremos a 1 la etapa inicial
If E1=0 and E2=0 and E3=0 and E4=0 and E5=0 then
  E1=1
Endif
'Si estando en la Etapa 1, y el final de carrera de cilindro dentro activo
'se activa el pulsador S1(Marcha) se inicia el ciclo pasando a la Etapa 2
If E1=1 and FC1=1 and S1=1 then
  E1=0
  E2=1
Endif
'Si se está en la Etapa 2 se hará un SET a la salida del indicador luminoso
If E2=1 and Temp=0 then
  L1=1
  DLY 3
  Temp=1
Endif
'El paso de la Etapa 2 a la etapa 3 será mediante una temporización.
If E2=1 and Temp=1 then
  E2=0
  E3=1
  Temp=0
Endif
'Si se está en la Etapa 3 se activa la salida del cilindro expendedor de
muelles
If E3=1 then CIL1=0
'Estando en la Etapa 3, y con el final de carrera de cilindro fuera
'si se pulsa paro se pasará a la Etapa 4
If E3=1 and FC2=1 and S2=0 then
  E3=0
  E4=1
Endif
'Si se está en la Etapa 4 se hará retroceder al cilindro.
If E4=1 then CIL1=1
'A la etapa 5 se llegará si estando en la Etapa 4 se activa el final de
'carrera de cilindro dentro.
If E4=1 and FC1=1 then
  E4=0
  E5=1
Endif
'En la Etapa 5 se apagará el indicador luminoso, terminando el ciclo
If E5=1 then L1=0

'De la etapa 5 se saldrá a la Etapa inicial E1 con un true,
'es decir, siempre. Se utiliza esta etapa para apagar el indicador luminoso
If E5=1 then
  E5=0
  E1=1
Endif

'Al finalizar el programa se salta al inicio para que este sea cíclico.
Goto *Inicio
```